
Worms-style scrolling engine
Introduction

One of the great things of Worms was that the environment was almost completely destructible. You could make holes and tunnels through the whole landscape. This make some nice battle techniques possible. However, I will write more about how you can program this kind of landscapes. As an extra, I thought it would be nice to write something about scrolling and maps that are larger then the screen. I don’t know if this would work in Qbasic, it could be too slow. It is anyway a quite good way to save memory for big pixel maps and other things (I originally found this technique in a font tutorial by Jakob Krarup).
Theory about destructible landscapes

Destructible landscapes are in theory just a matter of clearing the map buffer where it is hit by an explosion. The hard thing about this is that you need very much memory to save the whole map in pixels, because you can use tiles. For simplicity, I will first do it with one single map that is screen-size in Qbasic:

<code>

DIM map(32001)

</code>

Then, we don’t use this as our screenbuffer or something, but we have just 2 kind of pixels in it:

2 = Indestructible

1 = Destructible ground

0 = Air

Now when we write to screen with a loop, we check if a pixel is not zero, so if there is ground (doesn’t matter if it’s destructible or not now). If it is, we texture it. You can apply three textures: 
1. An air texture for the sky (this could be an image as well as a texture as well)

2. A texture for the destructible ground. Most people prefer some ground-colour for it. A brown, tiling texture is perfect.

3. And of course one for the indestructible ground (often, metal or some metal colour is used).

Now we draw it like this:

<code>

DEF SEG = VARSEG(map(0))

p& = 4

FOR x = 0 TO 319

  FOR y = 0 TO 199

    SELECT CASE PEEK(map(p&))

    CASE IS 0

      REM Apply cloud picture or texture

    CASE IS 1

      REM Texture with destructible ground texture

    CASE 2

      REM Texture with indestructible ground texture

    END SELECT

      p& = p& + 1

  NEXT

NEXT

</code>

Now we just use collision detection for our sprites. For the explosions we need to have a special map. We use it to say it if the ground is destructed, and if and how much damage is done to the character that it is hit by it (though this is not necessary as we can also just count how many pixels of the explosion are hitting  the character). A map would look like this:
	Explosion tile:
	Explosion damage map:

When the damage map hits a destructible pixel, it is removed. (I recommend round or random shapes instead of a map that leaves exact the same shape hole with every explosion). Ill call the white pixels damage pixels as they do damage. Instead of checking for every destructible pixel if it is hit by a damage pixel from the explosion map, you draw the explosion, and when on the place of a damage pixel is a destructible pixel, you erase it in the map. If you want to avoid DEF SEG (as it is not that fast), you should 1. Use a LUT to store  2. Use ASM. It is not easy but it is much faster and handles segments and such things better and faster then QB.

Here’s a little demo for you. I’m really sorry because it sucks so much, but you can still use it to look how the explosion sprite(really simple and ugly, sorry) and the explosion map are used. I’ve used here that the sprite is or drawed, or it doesn’t do any damage, so every drawed pixel is erased by the damage map, because it is drawed after it. If you don’t want this, you have to do this in the same order:

1. Put explosion sprite

2. (Eventually blit buffer to the screen)

3. (Loop the explosion animation if you have one; goto step one then)

4. Destroy the pixels with the damage map

5. Re-paint the background

6. (Eventually blit buffer to screen)

And here is my sucking demo. It doesn’t actually use a buffer, but just uses PSET and POINT, so it actually uses the buffer as a map. The way I would normally program it, is to make a buffer and PEEK and POKE to that. I could texture it too that way.
<code>

CLS

SCREEN 13

DEFINT A-Z

RANDOMIZE TIMER

FOR c = 16 TO 31

  OUT &H3C8, c

  OUT &H3C9, (c) / 31 * 63

  OUT &H3C9, (c) / 31 * 63

  OUT &H3C9, (c) / 31 * 63

NEXT

DIM ht(319)

FOR x = 0 TO 319

  ht(x) = RND * 199

NEXT

CONST smooth = 20

FOR b = 1 TO smooth

  FOR x = 0 TO 319

    ht(x) = (ht((x - 1 + 320) MOD 320) + ht(x) + ht((x + 1) MOD 320)) \ 3

  NEXT

NEXT

FOR x = 0 TO 319

  LINE (x, 199)-(x, 199 - ht(x))

NEXT

DIM map(-25 TO 25, -25 TO 25), expl(-25 TO 25, -25 TO 25)

FOR x = -25 TO 25

  x2 = x * x

  FOR y = -25 TO 25

    expl(x, y) = (31 - SQR(x2 + y * y) / 25 * 15)

    IF expl(x, y) < 17 THEN expl(x, y) = 0 ELSE expl(x, y) = expl(x, y) + RND * 10 - 5

    IF expl(x, y) < 17 THEN expl(x, y) = 0

    IF expl(x, y) > 31 THEN expl(x, y) = 31

    IF x2 + y * y < 625 THEN map(x, y) = 1

  NEXT

NEXT

DO UNTIL INKEY$ <> ""

  t! = TIMER + .2

  ex = RND * 319

  ey = RND * 199

  FOR xx = -25 TO 25

    FOR yy = -25 TO 25

      IF expl(xx, yy) > 15 THEN

        PSET (ex + xx, ey + yy), expl(xx, yy)

      END IF

    NEXT

  NEXT

  DO UNTIL TIMER >= t!: LOOP

  t! = TIMER + .3

  FOR xx = -25 TO 25

    FOR yy = -25 TO 25

      IF map(xx, yy) = 1 THEN

        rx = xx + ex: ry = yy + ey

        IF POINT(rx, ry) > 0 THEN PSET (rx, ry), 0

      END IF

    NEXT

  NEXT

  DO UNTIL TIMER >= t!: LOOP

LOOP

END
</code>

So. That is how you make a simple destructible terrain (I keep on typing destructable but Word helps me :D). But you only have a static, non-scrolling, boring map. Let’s make it bigger, so that we can scroll! First, though I will teach you how to actually use this map with holes.
Movement of the worms

To keep the worms on the right place their should be some kind of (primitive) collision detection. I suggest using an array that is the size of your worm (or other sprite ofcourse) where you hold which pixels are actually the worms and which pixels are empty. Then you only check the pixels which are not 0 (or another background colour). Kinda like the damage map and the explosion sprite. Here is the sprite, together with the collision detection array:

In this example you check only the pixels that are zero on the right map. You should only check the pixels that are new. New means that they are not already checked in previous collision detections.

Here, the red pixels are the ‘new’ pixels. You should do collision test for the whole black area every time that a worm has gone to a completely new position (teleportation or the first time that the worms are being placed somewhere). What to do when one of the pixels is not empty? Well, you should decide if the pixel is (relatively!) low enough to let your worms climb upon it or if it’s to high for your worm.

If h is lower then <a certain height>

What you should do also, is checking the pixels below the worm. 

Again, you should check only the red pixels. If they are empty, then move the worm down (it falls). Do this until:

1. It has hit some (destructible) ground. You could subtract some lives either, so that the worms could die if they fall from a high platform.

2. It is below the ground level. There are several things that you could do. You could let the worm disappear and act like it’s dead or either act just like the ground is destructible ground. So subtract some lives if the fall is high enough, and then just go on like normally.
Some kinds of projectiles
The most common weapons in worms are the grenade and the bazooka. They are relatively easy to code. You just have to do some simple collision detection for them. If the projectile hits a piece of ground, it should either explode(bazooka) or bounce(grenade). You can also make the explosion slightly bigger when the speed is bigger. The bouncing can be done by the easy way (invert speed on x or y-axis, dependant on where the collision was). You could also make some realistic kind of bouncing. Then you should have some way to calculate the normal on the point where the collision is and then use the reflection formula (originally used in raytracing, but also suitable for this purpose):
dot! = dot(Oldvector, Normalvector)

Newvector = Oldvector – 2 * dot! * Normalvector

All this vectors are 2d:

<code>

TYPE vector2d

  X AS SINGLE

  Y AS SINGLE

END TYPE

REM (Though you could use also integers with fixed point math)

</code>

Then, when we translate this from vector math to code:

<code>

TYPE vector2d

  X AS SINGLE

  Y AS SINGLE

END TYPE

REM (Though you could use also integers with fixed point math)

DIM Speed AS vector2d, Normal AS vector2d

dot! = dot(Speed.x, Speed.y, Normal.x, Normal.y)

Speed.x = Speed.x - 2 * dot! * Normal.x

Speed.y = Speed.y - 2 * dot! * Normal.y

</code>

For those who don’t know what the dot product is, here is the 2d version of it (there is also a 3d version of it, which is just the same but then with z component added too. Probably a 1d version also exists but I cant think of any possible purpose for this). Well, the code for the function is:

<code>

FUNCTION dot! (x1 AS SINGLE, y1 AS SINGLE, x2 AS SINGLE, y2 AS SINGLE)

dot! = x1 * x2 + y1 * y2

END FUNCTION

</code>

Big scrolling maps

Note: This is a manner of making big scrolling maps that I don’t know if it’s easy to make it run realtime, because I haven’t tried it. In theory this technique is quite easy, but when you actually use it is becomes much more complex due to the hard access to bits. To avoid this I suggest you use some lookuptables for the bit operations. I’ve explained how to make them in this chapter.

Bigger maps are cool to do. Remember that you have the textures stored not in the map, but separated? There is a reason for that. The map contains only if the pixel if destructible or empty. This is to save memory. We can, when use the binary mode, save big pictures in a smaller piece of memory. I suggest you’re using the way where you GET a piece if the screen loaded in an array. With this way you can save any portion of the screen with 256 colours. But when you have only two possible colours, the possibilities are much lower, so much less memory is used. When we use the same amount of memory(a screen size buffer), we can store 8 times larger pixel maps. Wow! But is has a downside. You can only store two kinds of pixels, so you have just:

· One texture for each kind of pixel. You could however make a tilemap, and paint the each pixel and lookup the texture at the tilemap, but this is somewhat hard.

· No un-destructible pixels. You can make these but then you can save ‘just’ four times a screen-size map in an array with screen-size. You just have air and destructible pixels else.

But it’s the only way I know, else then just using extreme much memory. Now you know about this, I will actually explain how this is done. If you don’t understand the bits and byte part, read the font tutorial by Jakob Krarup, or a bit tutorial by Ximmer. But first try to understand this, because it is not so hard, and I think I explained it not so bad.
Let’s take one pixel. A pixel is represented by one byte. A byte is built up from 8 bits. Bits have value from either 0 or 1. The computer uses decimal numbers to built up all decimal numbers. This is some really important stuff for in the rest of the tutorial.
Let me explain how it is done:

	1
	1
	1
	1
	1
	1
	1
	1

	128
	64
	32
	16
	8
	4
	2
	1


In this picture, all bits are ON (=1). This 8-bit integer(byte) represents a value. We find it by adding all places where an one is in the above row(so if the bit is ON), multiplied by the ‘weight’ of that bit(what’s standing below there). The weight for the nth bit is calculated by 2 ^ n. We start counting for n at zero, otherwise we would not get 1 at the begin.

So for the leftmost bit (I don’t call it ‘first’ because the rightmost bit is actually the first), it’s value is 1 * 128. Now, let’s add everything together(I left out the 1 *, because they are kinds useless):
128 + 64 + 32 + 16 + 8 + 4 + 2 + 1 = 255

Hey! 255, that is the maximum colour in SCREEN 13! Right. The colours in screen mode 13 are also built up out of 8 bits. 8 Bits can have any value between 0 and 255. But when we want only two colours, we can do some stuff and use bits for our colour, so either zero or one.

Then we have to have a way to convert the value(0 - 255) to the bit thing and back.

Well, we can say:

· If the value AND (I mean the AND operation now ;)) the weight of the bit is more then zero, then the bit is one. If it’s zero, the bit is off.
There is some mathematical explanation for this but I won’t bore you with it. Read the bit tutorial of you want to know what AND exactly does.
The code to paint 8 pixels from one byte value(0-255) is as follows:

<code>

FOR x = 0 TO 7

  IF value AND (2 ^ x) THEN PSET (x, 0), 15 ELSE PSET (x, 0), 0

NEXT

</code>

We can speed this code up with precalculating the values of n for every nth bit, but it’s smarter to remove the whole loop and do this:

<code>

IF value AND 1 THEN PSET (x, 0), 15 ELSE PSET (x, 0), 0

IF value AND 2 THEN PSET (x + 1, 0), 15 ELSE PSET (x + 1, 0), 0

IF value AND 4 THEN PSET (x + 2, 0), 15 ELSE PSET (x + 2, 0), 0

IF value AND 8 THEN PSET (x + 3, 0), 15 ELSE PSET (x + 3, 0), 0

IF value AND 16 THEN PSET (x + 4, 0), 15 ELSE PSET (x + 4, 0), 0

IF value AND 32 THEN PSET (x + 5, 0), 15 ELSE PSET (x + 5, 0), 0

IF value AND 64 THEN PSET (x + 6, 0), 15 ELSE PSET (x + 6, 0), 0

IF value AND 128 THEN PSET (x + 7, 0), 15 ELSE PSET (x + 7, 0), 0

</code>

In the game engine, you will need two LUT’s, one to convert from bits to a decimal value, that would look like:
<code>

DIM value(1, 1, 1, 1, 1, 1, 1, 1)

</code>

And another one that gives you the status of each nth bit with a certain value.

<code>

DIM bit(7, 255)

</code>

Where the second is the value of the byte you PEEK’d and the first is n (so you know the status of the nth bit). Of course this LUT’s are integer.

This is how we calculate out LUT’s:

<code>

DEFINT A-Z

DIM value(1, 1, 1, 1, 1, 1, 1, 1)

DIM bit(7, 255)

REM Calculate each bit of each value

FOR x = 0 to 255

  b0 = SGN(x AND 1): b1 = SGN(x AND 2): b2 = SGN(x AND 4): b3 = SGN(x AND 8):

  b4 = SGN(x AND 16): b5 = SGN(x AND 32): b6 = SGN(x AND 64): b7 = SGN(x AND 128)

  value(b0, b1, b2, b3, b4, b5, b6, b7) = x

  bit(0, x) = b0: bit(1, x) = b1: bit(2, x) = b2: bit(3, x) = b3

  bit(4, x) = b4: bit(5, x) = b5: bit(6, x) = b6: bit(7, x) = b7

NEXT

</code>

The SGN() keeps the value between 0 and 1 (or -1 if there is a negative value, but we don’t have that here). It returns 0 for 0, -1 for any negative number and 1 for every positive number.
The second LUT is to do the drawing, you put the value of a byte in it, and the number of the bit that you want to know, and then it returns the value of that bit.

The first is to do the invert, so this is to edit the map (for example for explosions). You put the 8 bits in it, and it returns the value of the byte that it would be. Then you just POKE that in the memory at the right offset.

Now when you want a scrolling map, you should use PEEK and POKE to get any value between 0 and 255, so you can look that up and change it easily with the LUT’s. I believe that you can make a bit lookuptable in FB to make things easier, but I’m not sure of that.

Now one thing that may be a bit hard is to find the offset from where too start.

The offset where you have to begin is:

(Width * camy + camx) \ 8

Then you start at the ((Width * camy + camx) AND 7)th bit. This is because at each offset, 8 pixels are stored (or 4 if you use more kinds of terrains). Then you paint the terrain, starting at the calculated offset (I just explained you how to do that), then you paint 8 pixels from the byte at that offset, then you increase the pointer by one, so it is pointing at the next offset. You do that 40 times for a screen width of 320, then you set the pointer to the start offset plus the screen width, and draw it all over again. Then you do that until you reached the bottom of the screen.
Conclusion

When I look back at this tutorial I realize that there was a bit too much theory. This is mainly because I hadn’t enough time and motivation to cover every little thing in detail, and write the code for it. This is possible for small things like particular effects, but if you would do it for an engine (and especially ones like this, that are pretty complex) you would get something like 30 pages. I prefer a ‘light’ tutorial, that is understandable and readable and one that still covers the very basic ideas behind it. Well, enough excuses about this tutorial which isn’t one of my best in my opinion. Please give me some feedback at basicallybest@live.nl. Greets,

Codemss
